def _maven_artifact(group, artifact, version, packaging = None, classifier = None, override_license_types = None, exclusions = None, neverlink = None):
    """
    Generates the data map for a Maven artifact given the available information about its coordinates.

    Args:
        group: *Required* The Maven artifact coordinate group name (ex: `"com.google.guava"`)
        artifact: *Required* The Maven artifact coordinate artifact name (ex: `"guava"`)
        version: *Required* The Maven artifact coordinate version name (ex: `"27.0-jre"`)

        packaging: *Optional* The Maven packaging specifier (ex: `"jar"`)
        classifier: *Optional* The Maven artifact classifier (ex: `"javadoc"`)
        override_license_types: *Optional* An array of Bazel license type strings to use for this artifact's rules (overrides autodetection) (ex: `["notify"]`)
        exclusions: *Optional* An array of exclusion objects to create exclusion specifiers for this artifact (ex: `maven.exclusion("junit", "junit")`)
        neverlink: *Optional* Determines if this artifact should be part of the runtime classpath.

    Output Schema:
        {
            "group": String
            "artifact": String
            "version": String
            "packaging": Optional String
            "classifier": Optional String
            "override_license_types": Optional Array of String
            "exclusions": Optional Array of exclusions (see below)
            "neverlink": Optional Boolean
        }
    """
    maven_artifact = {}
    maven_artifact["group"] = group
    maven_artifact["artifact"] = artifact
    maven_artifact["version"] = version

    if packaging != None:
        maven_artifact["packaging"] = packaging
    if classifier != None:
        maven_artifact["classifier"] = classifier
    if override_license_types != None:
        maven_artifact["override_license_types"] = override_license_types
    if exclusions != None:
        maven_artifact["exclusions"] = exclusions
    if neverlink != None:
        maven_artifact["neverlink"] = neverlink

    return maven_artifact